Discontinuous Distributions in Mechanics of Materials

J.E. Akin, Rice University

1. Introduction

The study of the mechanics of materials continues to change slowly. The student needs to learn about software support tools like general finite element systems, Maple, Matlab, and non-procedural case solvers such as TK Solver. Learning to use those tools takes away time that in the past was spent learning multiple ways for solving problems in mechanics of materials. Also, the focus of the study has changed. Today, mechanics of materials solutions are no longer an end in themselves. They are frequently used to validate the reasonableness of a complicated two- or three-dimensional finite element stress analysis.

Some classical mechanics of materials analysis methods have to be eliminated from the educational programs in order to make room for the more modern and more general analysis methods. Thus, the question is what analysis methods should be taught? One should teach those methods that are easily applied to statically indeterminate systems. For one-dimensional bars, shafts, and beams the finite element method offers the most power, followed by integration methods based on discontinuous distributions. The latter methods benefit from building on the student's basic knowledge of integral calculus, and free body diagrams. Here the use of discontinuous distributions for bars, shafts, and beams will be illustrated. Some of the example applications will also be implemented in TK Solver so their numerical use can also be illustrated.

2. Discontinuous Distribution Calculus

2.1 Definitions

The Macaulay distributions are denoted by a function, f(x), within triangular brackets that have an integer exponent, n: $< f(x) >^n$. The most commonly used ones in mechanics are:

Positive exponent distribution, $n > 0$:

$$< x - a >^n = \begin{bmatrix} 0 & for\ 0 < x < a \\ (x-a)^n & for\ a \le x < \infty \end{bmatrix}$$

Heaviside (unit step) distribution, $n = 0$:

$$H_a(x) = < x - a >^0 = \begin{bmatrix} 0 & for\ 0 < x < a \\ 1 & for\ a \le x < \infty \end{bmatrix}$$

Pole (unit pulse) distribution, $n = -1$:

$$P_a < x - a >^{-1} = \begin{bmatrix} 0 & for\ x \ne a \\ P_a & for\ x = a \end{bmatrix}$$

$$P_a = limit_{\varepsilon \to 0} \int_{a-\varepsilon/2}^{a+\varepsilon/2} \frac{P(x)}{\varepsilon} dx$$

Dipole (doublet pulse) distribution, $n = -2$:

$$M_a < x - a >^{-2} = \begin{bmatrix} 0 & for\ x \neq a \\ M_a & for\ x = a \end{bmatrix}$$

$$M_a = limit_{\varepsilon \to 0} \int_{a-\varepsilon}^{a} \frac{M(x)}{\varepsilon^2} dx - \int_{a}^{a+\varepsilon} \frac{M(x)}{\varepsilon^2} dx$$

2.2 Products

$$H_a H_a = H_a, \quad H_a H_b = H_b\ for\ a < b$$

2.3 Integrals

$$\int < x - a >^{\boldsymbol{n}} dx = \begin{cases} < x - a >^{n+1} & for\ n \leq 0 \\ \dfrac{< x - a >^{n+1}}{n+1} & for\ n > 0 \end{cases}$$

$$\int_{-\infty}^{x} H_a f(x) dx = H_a \int_{a}^{x} f(x) dx$$

$$\int H_a (x - a) dx = \int < x - a >^1 dx$$

$$\int < x - a >^n f(x) dx = \int H_a (x - a)^n f(x) dx = H_a \int_{a}^{x} (x - a)^n f(x) dx$$

$$\int H_b < x - a >^n dx = H_b \frac{< x - a >^{n+1}}{n+1} - H_b \frac{(b - a)^{n+1}}{n+1}\ for\ a < b$$

3. Axial bars

If you are learning to use only one or two techniques then you want to be sure that they can handle statically indeterminate systems. Both discontinuous distributions and finite elements can do that. Either approach requires the basic starting point of a correct “Free Body Diagram” (FBD) that displays both the loading and reactions.

The differential equation of equilibrium of an axially loaded linearly elastic bar, in terms of its displacement, u(x), is

-[E(x) A(x) {u(x)’ – α(x) ΔT(x)}]’ = p(x)

where E, α are the material’s elastic modulus and thermal expansion coefficient, A is the cross-sectional area, ΔT is the temperature increase, p is the axial load per unit length, and ( )’ = d( )/dx. Clearly, you must select the x-coordinate system relative to the bar to be studied. Usually, one end of the bar is chosen as the origin. Examine the free body diagram to see if either end has a prescribed displacement. Selecting such a point as the x-origin usually slightly simplifies the study, but it is not required.



As a first example, consider a statically indeterminate bar with constant E, A, α, and ΔT. The bar is fixed at both ends, so place the origin at the left end. Then the boundary conditions are $u(0) = 0 = u(L)$, and thus the FBD should show that it has two point reaction forces of R_0 and R_L at x = 0 and x = L, respectively. The bar is subjected to an axial point load P_a at $x = a < L$. The external load per unit length is easily given by three unit pulse discontinuous distributions as

$$p(x) = R_0\langle x\rangle^{-1} + P_a\langle x-a\rangle^{-1} + R_L\langle x-L\rangle^{-1}$$

Integrating this distribution once gives the resultant axial force, F(x), acting at each point along the full length of the bar:

$$F(x) = R_0\langle x\rangle^{0} + P_a\langle x-a\rangle^{0} + R_L\langle x-L\rangle^{0} + C,$$

where the constant of integration is always zero because all loads and reactions were included in p(x). That is, $F(0) = R_0 = R_0 + C$, so you normally omit C. The axial stress at a point is $\sigma(X) = F(x) / A(x)$. Next, you observe the closure (force equilibrium) condition that $F(L+) \equiv 0$ (or $\Sigma Fx = 0$):

$$R_0 + P_a + R_L = 0, \qquad \text{(Eq 1)}$$

This is the first equation involving the two unknown reactions. Since $-EA\,[u(x)' - \alpha\,\Delta T] = F(x)$, or $-u(x)' = F(x) / EA - \alpha\,\Delta T$ one more integration will give the axial displacement:

$$-u(x) = [R_0\langle x\rangle^{1} + P_a\langle x-a\rangle^{1} + R_L\langle x-L\rangle^{1}] / EA - \alpha\,\Delta T\, x + C_2$$

The first displacement boundary condition gives $u(0) = 0 = 0 - 0 + C_2$. The second displacement boundary condition gives

$$-u(L) = 0 = [R_0\, L + P_a(L-a) + 0] / EA - \alpha\,\Delta T\, L \qquad \text{(Eq 2)}$$

which is the second equation for the reactions. In general, the two unknown reactions are obtained by solving the simultaneous Eqs. 1 and 2. That is easily done numerically with TK Solver, or symbolically with Maple or similar software tools.

There are two common special cases:

a) $P_a = 0$ so only the thermal load is present which gives $R_0 = EA\,\alpha\,\Delta T = -R_L$ so $u(x) = [EA\,\alpha\,\Delta T\,\langle x\rangle^{1}]/EA - \alpha\,\Delta T\, x = 0$. There is no displacement, but there is a constant thermal force, R_0, and a constant compressive thermal stress, R_0 / A.

b) T = 0 so there is only the point load. That gives the reactions of $R_0 = -P_a(L-a) / L$ and $R_L = -P_a(a / L)$, as expected. There are two regions of constant stress in the bar and its deflection is

$$-u(x) = [-P_a(L-a) <x>^1 / L + P_a <x-a>^1 - P_a(a / L) <x-L>^1] / EA$$

$$-u(x) = [-(L-a)\, x/L + <x-a>^1]\, P_a / EA$$

which of course vanishes at x = 0 and x = L, as required.

Combining these two cases gives the general solution

$$-u(x) = \{[EA\, \alpha\, \Delta T - P_a(L-a) / L]\, x + P_a <x-a>^1\} / EA - \alpha\, \Delta T\, x.$$

Note that the right end point term involving $<x-L>^n$ need not appear explicitly in the final expression for the displacement, u(x). However, its inclusion is necessary, in p(x), to obtain the reactions which always appear in u(x). The above examples, and others including distributed loads per unit length, are illustrated in the following sections on software implementations.

A more general form of this example would be to assume that the end displacements are given non-zero values of u_0 and u_L, respectively. Then one can include support settlement cases rather than just fixed ends.

4. TK Solver example implementations

The TK Solver case solving software has been widely utilized in engineering. The theory behind case solvers is covered in [1, 3], and descriptions of the general capabilities of TK Solver are found at [6]. The text by Norton [4] covers many mechanical engineering applications including statics, dynamics, mechanics of materials, etc. solve with TK Solver.

The rules for TK Solver can be in any order and thus are non-procedural, unlike C, Fortran, Visual Basic, etc. The unknown can be on either side of the equals (=) symbol. Mixed units can be used so long as the unit selected exists as a symbol in the Units Sheet. When adding new rules you must give the units symbol (in column 5 of the Variables Sheet) before solving the first time. A function sfn (x, a, n) to evaluate the logic for $< x – a >^n$ was defined in the Function Sheet to include the range of applications common to mechanics of materials. Its name is short for **s**ingular **f**unctio**n** to be consistent with the notation of [4]. Two examples of defining such procedures are given in Figure 1.

This implementation is for the first axial bar example, plus the addition of a constant line load over part or all of the length. The end at x = 0 is always fixed, but the far end displacement can be assigned a non-zero value. The rules are given in Figure 2. Those rules that have the position, x, as an argument can be evaluated at any point, and produce line plots when computed as a List Solve. The equilibrium rule, $F(L^+) = 0$, and the displacement boundary condition, u(L) = u_L, provide the data necessary to recover the two reaction forces.

PROCEDURE: SFN

Comment:	Selector
Parameter Variables:	
Input Variables:	x,a,n
Output Variables:	y

St	Statement
	; the following are procedural statements
	y = 0
	if n = -2 then y = DOUBLET (x,a,1,1e-4)
	if n = -1 then y = PULSE (x,a,1)
	if n = 0 then y = STEPF (x,a,1)
	if n > 0 then y = GENERAL (x,a,n)

PROCEDURE: GENERAL

Comment:	x>a, n>0 Singularity Function
Parameter Variables:	
Input Variables:	x,a,n
Output Variables:	y

St	Statement
	if x <= a then y = 0 else y = (x - a)^n

Figure 1 Functions to evaluate discontinuous distributions

Rules

<table>
<tr><th>Status</th><th>Rule</th></tr>
<tr><td>Comment</td><td>; |---- b --->wwwwwwwwwwwwwwwwwwww   ; Indeterminate Axial Bar with Point, Uniform & Thermal Loads</td></tr>
<tr><td>Comment</td><td>; R0 ^------- a --------->Pa ^   RL ^</td></tr>
<tr><td>Comment</td><td>; |------------ L --------------------------------->   ; J.E. Akin, 2006</td></tr>
<tr><td>Comment</td><td>; SFN (x,a,n) == <x - a> ^n singularity function</td></tr>
<tr><td>Satisfied</td><td>p   = R0*SFN(x,0,-1) + Pa*SFN(x,a,-1) + w*SFN(x,b,0) + RL*SFN(x,L,-1) ; load per unit length</td></tr>
<tr><td>Satisfied</td><td>F   = R0*SFN(x,0,0) + Pa*SFN(x,a,0) + w*SFN(x,b,1) + RL*SFN(x,L,0)   ; axial force</td></tr>
<tr><td>Satisfied</td><td>0   = R0*SFN(L,0,0) + Pa*SFN(L,a,0) + w*SFN(L,b,1) + RL*SFN(L,L,0)   ; EQ 1: F(L+) = 0, or ΣFx = 0, equilibrium</td></tr>
<tr><td>Satisfied</td><td>u   = -(R0*SFN(x,0,1) + Pa*SFN(x,a,1) + w*SFN(x,b,2)/2) / EA - αTx   ; axial displacement, RL*SFN(x,L,1) == 0</td></tr>
<tr><td>Satisfied</td><td>u_L = -(R0*SFN(L,0,1) + Pa*SFN(L,a,1) + w*SFN(L,b,2)/2) / EA - αTL   ; EQ 2: u(L) = u_L,   RL*SFN(L,L,1) == 0</td></tr>
<tr><td></td><td></td></tr>
<tr><td>Comment</td><td>; find stress, etc</td></tr>
<tr><td>Satisfied</td><td>Applied = w*(L-b) + Pa   ; Total applied load</td></tr>
<tr><td>Satisfied</td><td>Reactions = R0 + RL   ; Reactions total</td></tr>
<tr><td>Satisfied</td><td>Sxx = F / A   ; Normal stress at x</td></tr>
<tr><td>Satisfied</td><td>EAbyL = EA / L   ; Axial stiffness</td></tr>
<tr><td>Satisfied</td><td>EA = E * A   ; Product</td></tr>
<tr><td>Satisfied</td><td>αTx = α * T * x   ; Thermal expansion</td></tr>
<tr><td>Satisfied</td><td>αTL = α * T * L   ; Max thermal expansion</td></tr>
</table>

Figure 2 Rules for axial bar with point loads and partial uniform load

The Variable Sheet is seen in Figure 3. The particular input data displayed there are for a point load acting near the middle of the bar. Three variables (Reactions, Applied, and u) have been displayed in metric units just to note that it is easy to switch them. Just enter the new symbol in the units column and the numerical value is instantly converted.

The basic list plots for those data are given in Figure 4. They include the load per unit length, axial force, axial stress, and axial displacement, respectively from top to bottom. In Figure 5 similar plots are given for a bar with a constant load per unit length of w = 100 lb/in and having both ends fixed. The top half of the bar is in tension while the bottom half has a mirror image compressive stress distribution.

Variables

Status	Input	Name	Output	Unit	Comment
					Indeterminate Axial Bar with Point, Uniform & Thermal Loads
					; |---- b --->wwwwwwwwwwwwwwwwwwwwww
					; R0 ^------- a --------->Pa ^ RL ^
					; |------------ L --------------------------------> ; J.E. Akin, 2006
					INSTRUCTIONS
					1 - Input bar dimensions, loads, and support movement
					2 - Direct Solve to find R0, RL
					3 - Move R0 and RL to input values (put an I in status column)
					4 - Do a LIST SOLVE (F10) to generate the plots pulldown
					Bar Geometry, Load, and Temperature Increase
	12	L		in	length, and distance to reaction RL
	4	a		in	distance to point load, 0 <= a <= L
	100	Pa		lb	point load value at 'a', -, 0, +
	.1	pt_size		in	length of point loads, say L/100, for Sxx only
	6	b		in	point where uniform load begins, 0 <= b < L
	0	w		lb/in	applied uniform load magnitude, -, 0, +
	0	u_L		in	settlement at x=L (at RL), usually zero
	3.00E7	E		psi	Young's elastic modulus, and
		EAbyL	5000000	lb /in	axial stiffness, EA / L, , or
	2	A		in^2	cross-sectional area, , or
		EA	6E7	lb	E * A, , are input
	6.7E-6	α		1/F	thermal expansion coefficient
	0	T		F	temperature increase, -, 0, +
					Reactions, misc.
		R0	-66.66667	lb	left reaction
		RL	-33.33333	lb	right reaction
		Reactions	-444.82	N	reactions total
		Applied	444.82	N	total applied external load
		αTL	0	in	max hermal expansion
					Bar functions of position (x). List solve for plots.
L	8	x		in	distance along bar for calculation
L		p	0	lb/in	load per unit length at x
L		F	33.33333	lb	axial force at x
L		Sxx	16.66667	psi	flexural stress at x
L		u	5.644E-5	mm	axial displacement at x
L		αTx	0	in	thermal expansion

Figure 3 Variables for axial bar with point loads and partial uniform load



Figure 4 Graphs for fixed-fixed bar with a point load



Figure 5 Graphs of a fixed-fixed bar with constant load per unit length

To model a bar hanging under its on weight you view w as the weight per unit length, set R0 = wL, the total weight, so RL is set to zero. A Direct Solve yields the end displacement of u_L = 1.2e-3 inches, which corresponds to the analytic free end displacement of umax = (wL / 2) L / EA. Now the linear stress distribution is all tension, being zero at the free bottom end and maximum at the top (x = 0) support. The corresponding list plots are seen in Figure 6.



Figure 6 Graphs for fixed-free axial bar hanging under its on weight

The following section will address the torsion of circular shafts. Then the more common use of discontinuous distributions for beam deflections will be reviewed. Finally, the process for changes in materials or part sizes will be outlined, even though they are better suited to finite element analysis.

5. *Axial shaft in torsion*

The problem of a circular bar subjected to axial torsional couples is basically the same as the above problem expressed with different symbol meanings. The differential equation of torsional equilibrium, in terms of its angular displacement (rotation), θ(x), is

$$-[G(x)\ J(x)\ \theta(x)'\ ]' = t(x)$$

where G is the material's shear modulus, J is the polar moment of inertia of the cross-sectional area, t is the axial torque per unit length, and ( )' = d( )/dx. The classic example is a bar with constant GJ, fixed at x = 0, and subjected to a torque T_L at its free end. The torque per unit length is

$$t(x) = T_0 <x>^{-1} + T_L <x-L>^{-1}$$

where T_0 is the unknown reaction torque. Integration gives the axial torque,

$$T(x) = T_0 <x>^0 + T_L <x-L>^0.$$

The closure of $T(L^+) \equiv 0$ (or torque equilibrium $\Sigma\ T_x = 0$) gives $T_0 + T_L = 0$. The wall reaction torque is $T_0 = -T_L$. Integrating -GJ θ(x)' = T(x) yields the rotation

$$-GJ\ \theta(x) = -T_L <x>^1 + T_L <x-L>^1 = -T_L\ (x + <x-L>^1).$$

The maximum rotation is $\theta(L) = T_L\ L\ /GJ$, as expected.

As a second example, consider the shaft to have both ends fixed and a linear distributed torque per unit length given by $[T_a + (T_b - T_a)\ x\ /\ L]$. Including the two indeterminate reaction torques you begin with

$$t(x) = T_0 <x>^{-1} + H_0\ [T_a + (T_b - T_a)\ x\ /\ L] + T_L <x-L>^{-1}$$

which yields a torque along the length of the bar of

$$T(x) = T_0 <x>^0 + H_0\ [T_a\ x + (T_b - T_a)\ x^2\ /\ 2L] + T_L <x-L>^0.$$

The maximum shear stress, for a circular bar, is $\tau = T(x)\ r\ /\ J$. Equilibrium closure gives

$$T(L^+) = 0 = T_0 + [T_a\ L + (T_b - T_a)\ L\ /\ 2] + T_L\ ,\ \text{or}\quad T_0 + T_L + (T_a + T_b\ )\ /\ 2 = 0\ . \qquad \text{EQ 1}$$

For constant GJ the rotation becomes

$$-GJ\ \theta(x) = T_0 <x>^1 + [T_a\ x^2\ /\ 2L + (T_b - T_a)\ x^3\ /\ 6L] + T_L <x-L>^1.$$

Enforcing the second boundary condition, $\theta(L) = 0$, reduces this to

$0 = T_0 L + [T_a L / 2 + (T_b - T_a) L^2 / 6] + 0$, or $0 = T_0 + T_a / 2 + (T_b - T_a) L / 6$. EQ 2

Combining equations 1 and 2 gives the expressions for the reactions T_0 and T_L. Substituting their expressions gives the final shaft rotation

$$-GJ\, \theta(x) = -x\, [T_a / 2 + (T_b - T_a) L / 6] + [T_a x^2 / 2L + (T_b - T_a) x^3 / 6L] + 0$$

$$-GJ\, \theta(x) = (x^2 / L - x)\, T_a / 2 + (x^3 / L - x L)(T_b - T_a) / 6.$$

This example, and others including distributed torques per unit length , is illustrated in the later sections on software implementations.

5.1 TK Solver shaft models

A similar set of TK rules for shaft torsion are given in Figure 7 .For the classic problem of two equal and opposite end torques, the rotation is a linear function of position as given by the above equation and as shown in Figure 8. That was accomplished by specifying a zero far end reaction torque and solving for the non-zero end rotation.

Rules

Status	Rule
Comment	; |tttttttttttttttttttttttttttttttttttttttttttttt ; Indeterminate Shaft with Point & Linear Distributed Torque Loads
Comment	; R0 ^------- p --------->Tp^ RL ^
Comment	; |------------ L ------------------------------> ; J.E. Akin, 2006
Comment	; SFN (x,a,n) == <x - a>^n singularity function
Satisfied	t = R0*SFN(x,0,-1) + Tp*SFN(x,p,-1) + Line_t + RL*SFN(x,L,-1) ; torque per unit length
Satisfied	T = R0*SFN(x,0,0) + Tp*SFN(x,p,0) + Line_T + RL*SFN(x,L,0) ; axial force
Satisfied	0 = R0*SFN(L,0,0) + Tp*SFN(L,p,0) + Line_TL + RL*SFN(L,L,0) ; EQ 1: T(L+) = 0, or ΣMx = 0, equilibrium
Satisfied	u = -(R0*SFN(x,0,1) + Tp*SFN(x,p,1) + Line_u) / GJ ; axial rotation, RL*SFN(x,L,1) == 0
Satisfied	u_L = -(R0*SFN(L,0,1) + Tp*SFN(L,p,1) + Line_uL) / GJ ; EQ 2: u(L) = u_L, RL*SFN(L,L,1) == 0
Comment	; Distributed line torque definitions (comment out old ones)
Satisfied	Line_t = (Ta + (Tb - Ta)*x / L) ; rectangular line torque defined
Satisfied	Line_T = (Ta*x + (Tb - Ta)*x^2 / 2 / L) ; rectangular line torque term
Satisfied	Line_u = (Ta*x^2 /2 + (Tb - Ta)*x^3 / 6 / L) ; rectangular line torque term
Satisfied	Line_TL = (Ta*L + (Tb - Ta)*L / 2) ; rectangular line torque term at x = L
Satisfied	Line_uL = (Ta*L^2 /2 + (Tb - Ta)*L^2 / 6) ; rectangular line torque term at x = L
Comment	; find stress, etc
Satisfied	Applied = (Tb + Ta)*L / 2 + Tp ; Total applied torque
Satisfied	Reactions = R0 + RL ; Reactions total
Satisfied	Tau= T * r / J ; Shear stress at x
Satisfied	GJbyL = GJ / L ; Torsional stiffness
Satisfied	GJ = G * J ; Product
Satisfied	J = pi()*r^4 /2 ; Polar moment of inertia
Satisfied	A = pi()*r^2 ; Area

Figure 7 Rules for shaft with torques at given points



Figure 8 Shaft with two equal and opposite end torques

One difference in the torsional rule set is that the distributed torque per unit length has been defined by a variable Line_t so that its contribution to t(x) will be easy to modify. Here it is set for a trapezoidal variation in the distributed line torque. The next example applies that trapezoidal distributed torque, and fixes the rotation at each end of the shaft. No external point torque (T_p) is applied. The distributed and resultant torques are shown in Figure 9, while the shaft rotation is given Figure 10. The corresponding input and output variables are given in the Variable Sheet of Figure 11.



Figure 9 Fixed-fixed shaft with trapezoidal distributed torque per unit length



Figure 10 Rotations of fixed-fixed shaft with variable distributed torque

Variables

<table>
<tr><th>Status</th><th>Input</th><th>Name</th><th>Output</th><th>Unit</th><th>Comment</th></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>; |ttttttttttttttttttttttttttttttttttttttttttttt   ; Indeterminate Shaft with Point &</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>; R0 ^------- p --------->Tp^   RL ^   ; Linear Distributed Torque Loads</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>; |------------ L ----------------------------->   ; J.E. Akin, 2007</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>INSTRUCTIONS</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>1 - Input shaft dimensions, torques, and support rotation</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>2 - Direct Solve to find R0, RL</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>3 - Move R0 and RL to input values (put an I in status column)</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>4 - Do a LIST SOLVE (F10) to generate the plots pulldown</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td></td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>Shaft Geometry,and Torques</td></tr>
<tr><td></td><td>12</td><td>L</td><td></td><td>in</td><td>length, and distance to reaction RL</td></tr>
<tr><td></td><td>0</td><td>Tp</td><td></td><td>in-lb</td><td>point torque at x = p</td></tr>
<tr><td></td><td>0</td><td>p</td><td></td><td>in</td><td>location of point torque</td></tr>
<tr><td></td><td>50</td><td>Ta</td><td></td><td>in-lb/in</td><td>distributed torque at x = 0</td></tr>
<tr><td></td><td>150</td><td>Tb</td><td></td><td>in-lb/in</td><td>distributed torque at x = L</td></tr>
<tr><td></td><td>0</td><td>u_L</td><td></td><td>in</td><td>rotation at x=L (at RL), usually zero</td></tr>
<tr><td></td><td>1.15E7</td><td>G</td><td></td><td>psi</td><td>material shear modulus   is input</td></tr>
<tr><td></td><td></td><td>GJbyL</td><td>23997541</td><td>in-lb</td><td>torsional stiffness, GJ / L   or</td></tr>
<tr><td></td><td></td><td>GJ</td><td>2.8797E8</td><td>lb-in^2</td><td>product G * J</td></tr>
<tr><td></td><td>2</td><td>r</td><td></td><td>in</td><td>radius   is input</td></tr>
<tr><td></td><td></td><td>A</td><td>12.56637</td><td>in^2</td><td>cross-sectional area,   or</td></tr>
<tr><td></td><td></td><td>J</td><td>25.13274</td><td>in^4</td><td>polar moment of inertia</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td></td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>Reactions</td></tr>
<tr><td></td><td></td><td>R0</td><td>-500</td><td>in-lb</td><td>reaction torque at x = 0</td></tr>
<tr><td></td><td></td><td>RL</td><td>-700</td><td>in-lb</td><td>reaction torque at x = L</td></tr>
<tr><td></td><td></td><td>Reactions</td><td>-1200</td><td>in-lb</td><td>sum of reactions</td></tr>
<tr><td></td><td></td><td>Applied</td><td>1200</td><td>in-lb</td><td>sum of applied torques</td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td></td></tr>
<tr><td></td><td></td><td></td><td></td><td></td><td>Bar functions of position (x). List solve for plots.</td></tr>
<tr><td>L</td><td>8</td><td>x</td><td></td><td>in</td><td>distance along bar for calculation</td></tr>
<tr><td>L</td><td></td><td>t</td><td>116.6667</td><td>in-lb/in</td><td>distributed torque at point x</td></tr>
<tr><td>L</td><td></td><td>T</td><td>166.6667</td><td>in-lb</td><td>torque at point x</td></tr>
<tr><td>L</td><td></td><td>Tau</td><td>13.26291</td><td>psi</td><td>max shear stress at point x</td></tr>
<tr><td>L</td><td></td><td>u</td><td>5.865E-6</td><td>rad</td><td>rotation at point x</td></tr>
</table>

Figure 11 Variables for point and distributed shaft torques

6. Axial beam in bending

Consider a straight, elastic beam undergoing small transverse displacements, v(x), when subjected to a transverse load per unit length of q(x). The governing differential equation of equilibrium, assuming no thermal moment due to temperature change through the depth, is

$$[E(x)\, I(x)\, v''(x)]'' = q(x)$$

where E is the elastic modulus, $I = \int y^2\, dA$ is the moment of inertia of the cross-sectional area, A, and ( )' = d( )/dx. It is also assumed that the slope of the deflection is small, $v' << 1$. Integrating once gives the transverse shear force V(x) as

$$[E(x)\, I(x)\, v''(x)]' = dM(x) / dx = V(x)$$

followed by the bending moment-curvature relation

$$E(x)\, I(x)\, v''(x) = M(x)$$

the slope

$$v'(x) = \theta(x) = \int [M(x) / E(x)\, I(x)]\, dx$$

and finally the deflection

$$v(x) = \int \theta(x)\, dx.$$

In most problems the product of EI is a constant. Then you get the common terminology for the beam:

Load:	$EI\, v'''' = q(x)$
Shear:	$EI\, v''' = V(x)$
Moment:	$EI\, v'' = M(x)$
Slope:	$v' = \theta(x)$, and
Deflection:	$v(x)$.

Of course, these will involve constants of integration to be determined from the displacement and slope boundary conditions in conjunction with the static equations of equilibrium:

$$\Sigma F_y = 0,\ \ \Sigma M_p = 0,$$

where p is any point in the x-y plane.

6.1 Statically determinate example

As an example application of discontinuous distributions applied to beams consider a simply supported beam with a constant EI, a length L, with a downward point load, P_a, at x=a, a pure couple $-C_b$ at x=b, and a downward uniform load per unit length, w, running from point x=c to the end of the beam. A FBD shows upward reaction forces at the two ends of R_0 and R_L, respectively. The load per unit length is

$$[EI\ v'']'' = q(x) = R_0\langle x\rangle^{-1} - P_a\langle x-a\rangle^{-1} - C_b\langle x-b\rangle^{-2} - w\langle x-c\rangle^{0} + R_L\langle x-L\rangle^{-1}.$$

The transverse shear force is

$$[EI\ v'']' = V(x) = R_0\langle x\rangle^{0} - P_a\langle x-a\rangle^{0} - C_b\langle x-b\rangle^{-1} - w\langle x-c\rangle^{1} + R_L\langle x-L\rangle^{0},$$

so the shear equilibrium closure gives

$$V(L^+) = 0 = R_0 - P_a - 0 - w(x-c) + R_L\ . \qquad \text{EQ 1}$$

Integrating again for the bending moment

$$EI\ v'' = M(x) = R_0\langle x\rangle^{1} - P_a\langle x-a\rangle^{1} - C_b\langle x-b\rangle^{0} - w\langle x-c\rangle^{2} / 2 + R_L\langle x-L\rangle^{1},$$

and moment equilibrium closure (sum of the moments about x=L) gives

$$M(L^+) = 0 = R_0\, L - P_a(L-a) - C_b - w(L-c)^2 / 2 \qquad \text{EQ 2}$$

In this case, the two closure equations allow for the solution of the two reactions, R_0 and R_L. The slope becomes:

$$v'(x) = \theta(x) = [R_0\langle x\rangle^{2} /2 - P_a\langle x-a\rangle^{2} /2 - C_b\langle x-b\rangle^{1} - w\langle x-c\rangle^{3} / 2 / 3 + R_L\langle x-L\rangle^{2} /2 + c_1] / EI$$

and finally the deflection is

$$v(x) = [R_0\langle x\rangle^{3} /6 - P_a\langle x-a\rangle^{3} /6 - C_b\langle x-b\rangle^{2} /2 - w\langle x-c\rangle^{4} / 24 + R_L\langle x-L\rangle^{2} /6 + c_1\, x] / EI + c_2.$$

The displacement boundary condition $v(0) = 0$ gives $c_2 = 0$. Likewise the displacement boundary condition $v(L) = 0$ yields the remaining constant, c_1, from

$$0 = R_0L^3 /6 - P_a(L-a)^3 /6 - C_b(L-b)^2 /2 - w(L-c)^4 / 24 + 0 + c_1\, L\ .$$

$$- c_1 = R_0L^2 /6 - P_a(L-a)^3 /6\, L - C_b(L-b)^2 /2\, L - w(L-c)^4 / 24\, L.$$

6.2 Statically indeterminate example

As an indeterminate example, to be implemented in TK Solver, consider a beam on three roller supports, one at each end and one at $x = b < L$. The beam is loaded by a downward uniform load over part of the span from $x = a$ to the end, $x = L$. The reactions, R1, R2, and R3 are assumed to be upward in the FBD. A sketch of the problem and the TK rules are given in Figure 12, while a sample Variable Sheet is given in Figure 13.

The loading per unit length is

$$q(x) = R_1 \langle x\rangle^{-1} - w \langle x - a\rangle^{0} + R_2 \langle x-b\rangle^{-1} + R_3 \langle x-L\rangle^{-1}$$


```
Rule
;      |---------------- a ------------->wwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwwww      ; Example 4-7, Machine Design, R. Norton, 1994.
;  R1 ^---------------------- b ------------------------------------------------->R2 ^                        R3 ^      ; Redundantly-Supported, Indeterminate Beam
;      |------------------------------------------------ L ------------------------------------------------------------>      ; Modified E. Akin, 2006

q  = R1*SFN(x,0,-1) - w*SFN(x,a,0) + R2*SFN(x,b,-1) + R3*SFN(x,L,-1)                          ; load
V  = R1*SFN(x,0,0)  - w*SFN(x,a,1) + R2*SFN(x,b,0)  + R3*SFN(x,L,0) + C1                       ; transverse shear
0  = R1*SFN(L,0,0)   - w*SFN(L,a,1) + R2*SFN(L,b,0)  + R3*SFN(L,L,0) + C1                      ; V(L+) = 0, or ΣFy = 0, equilibrium        EQ 1
C1 = 0                                                                                         ; Since V(0) = R1                           EQ 2
M = R1*SFN(x,0,1) - w*SFN(x,a,2)/2 + R2*SFN(x,b,1) + R3*SFN(x,L,1) + C1*x + C2                 ; bending moment
0  = R1*SFN(L,0,1) - w*SFN(L,a,2)/2 + R2*SFN(L,b,1) + R3*SFN(L,L,1) + C1*L  + C2               ;  M(L+) = 0, or ΣM = 0, equilibrium        EQ 3
C2 = 0                                                                                         ; Since M(0) = 0 by inspection              EQ 4
theta = (R1*SFN(x,0,2) /2   - w*SFN(x,a,3)/2/3    + R2*SFN(x,b,2) /2    + R3*SFN(x,L,2)/2   + C1*x^2/2    + C2*x       + C3) / EI          ; slope
y      = (R1*SFN(x,0,3) /2/3 - w*SFN(x,a,4)/2/3/4 + R2*SFN(x,b,3) /2/3 + R3*SFN(x,L,3)/2/3 + C1*x^3/2/3 + C2*x^2/2 + C3*x + C4) /EI  ; deflection
y_L  = (R1*SFN(L,0,3) /2/3 - w*SFN(L,a,4)/2/3/4 + R2*SFN(L,b,3) /2/3 + R3*SFN(L,L,3)/2/3 + C1*L^3/2/3 + C2*L^2/2 + C3*L + C4) /EI  ;y(L)=0 if no settlement  EQ 5
y_b  = (R1*SFN(b,0,3) /2/3 - w*SFN(b,a,4)/2/3/4 + R2*SFN(b,b,3) /2/3 + R3*SFN(b,L,3)/2/3 + C1*b^3/2/3 + C2*b^2/2 + C3*b + C4) /EI ; y(b)=0 if no settlement  EQ 6
C4 = y_0                                                                                                                          ; y(0)=0 if no settlement  EQ 7
                ;  There are 7 unknowns: 4 constants of integtation plus 3 reactions

; find stress, weight, etc
Applied = w*(L-a)                                    ; Total applied load
Reactions = R1 + R2 + R3                             ; Total of reactions
Sxx = M * c / I                                      ; Flexural stress at x
Sxy = 1.5 * V / width / depth                        ; Shear stress at x
Syy = (R1*SFN(x,0,-1) + R2*SFN(x,b,-1) + R3*SFN(x,L,-1) ) /pt_size / width - w*SFN(x,a,0) / width   ; transverse stress at x
Svm = sqrt (Sxx^2 + Syy^2 + 6*Sxy^2) * sqrt(0.5)    ; Von Mises stress at x
c = depth / 2
I = width * depth^3 /12
EI=E*I
```

Figure 12 Rules for three point supported beam with partial uniform load

The transverse shear force is

$$V(x) = R_1 \langle x\rangle^0 - w \langle x - a\rangle^1 + R_2 \langle x\text{-}b\rangle^0 + R_3 \langle x\text{-}L\rangle^0$$

The force equilibrium closure is

$$V(L^+) = 0 = R_1 - w\,(L - a) + R_2 + R_3\,. \qquad \text{EQ 1}$$

The bending moment becomes

$$M(x) = R_1 \langle x\rangle^1 - w \langle x - a\rangle^2 / 2 + R_2 \langle x\text{-}b\rangle^1 + R_3 \langle x\text{-}L\rangle^1$$

so that the moment equilibrium closure requires

$$M(L^+) = 0 = R_1\, L - w\,(L - a)^2 / 2 + R_2\,(L\text{-}b) + 0 \qquad \text{EQ 2}$$

You could write other moment equilibrium equations, such as at x = 0 and x = b. That would add redundant equations, which is often useful in TK Solver, but is not necessary here. The slope is

$$v'(x) = [R_1 \langle x\rangle^2 / 2 - w \langle x - a\rangle^3 / 2 / 3 + R_2 \langle x\text{-}b\rangle^2 / 2 + R_3 \langle x\text{-}L\rangle^2 /2 + c_1] / EI$$

and the deflection becomes

$$v(x) = [R_1 \langle x\rangle^3 / 6 - w \langle x - a\rangle^4 / 24 + R_2 \langle x-b\rangle^3 / 6 + R_3 \langle x-L\rangle^3 / 6 + c_1 x] / EI + c_2$$

There are three displacement boundary conditions to be combined with the above two closure conditions to allow you to find the five unknowns (R_1, R_2, R_3, c_1, and c_2).

Input	Name	Output	Unit	Comment
				Example 4-7 in Machine Design by R Norton 1994
				Redundantly-Supported, Indeterminate Beam. Modified E. Akin, 2007
				Beam Geometry and Loads
10	L		in	length, and distance to reaction R3
1	a		in	distance to load w
6	b		in	distance to reaction R2
6	x		in	distance along beam for calculation
500	w		lb/in	applied load step magnitude
3.00E7	E		psi	Young's modulus
.080	I		in^4	beam area moment of inertia
	EI	2.40E6		product of E and I
	c	.6	in	distance to outer fiber
1.2	depth		in	depth of reectangular section
	width	.555555556	in	width of reectangular section
	weight	1.86666667	lb	weight of beam
.28	dens		lb/in^3	density of material
.1	pt_size		in	length of point loads, say L/100, for Syy & Svm only
0	y_0		in	settlement at x=0 (at R1)
0	y_b		in	settlement at x=b (at R2)
0	y_L		in	settlement at x=L (at R3)
	Applied	4500	lb	total applied load
				Reactions
	Reactions	4500	lb	sum of reactions
762.326389	R1		lb	left (a guess value is required)
3156.68403	R2		lb	center (a guess value is required)
	R3	580.989583	lb	right
				Beam Functions
	q	2656.68403	lb	loading function at point x
	V	1419.01042	lb	shear function at point x
	M	-139.670139	ft-lb	moment function at point x
	theta	.000375579	rad	slope function at point x
	y	0.0000	in	deflection function at point x
	Sxx	-12570.3125	psi	flexural stress at x
	Sxy	3192.77344	psi	shear stress at x
	Syy	55920.3125	psi	transverse normal stress at x
	Svm	40903.893	psi	Von Mises failure criteria at x
				; |--------- a ------>wwwwwwwwwwwwwwwwwwwwwwwwwwwwwwww
				; R1 ^-------------- b ---------------------->R2 ^ R3 ^
				; |------------------------------------- L -------------------------------->

Figure 13 Variables for three point supported beam with partial uniform load

The three support points could have non-zero settlements, as allowed for in the TK implementation. For the input choices in the Variable Sheet the shear, moment, slope and deflection are graphed in Figure 14. Likewise, one can compute the flexural stress, M c / I, the shear stress, 1.5 V / A, an approximate transverse (y) normal stress, q(x)/width, and a failure criteria by Von Mises. These are graphed in Figure 15. The y-normal stress is ignored in beam theory, but is quite high due to the assumed point reaction forces.



Figure 14 Example shear, moment, slope and deflection of three point beam



Figure 15 Corresponding stresses for example three point beam

References

1. Akin, J. E.: Computer Assisted Mechanical Design, Prentice-Hall, Englewood Cliffs, NJ, 1990.
2. Craig, R.R., Mechanics of Materials, 2^{nd} Ed, John Wiley, New York, NY, 2000.
3. Konopasek, M.; Jayaraman, S.: The TK Solver Book, McGraw-Hill, New York, NY, 1984.
4. Norton, R.L., Machine Design: An Integrated Approach, Prentice Hall, Upper Saddle River, NJ, 1996.
5. Popov, E.P., Engineering Mechanics of Solids, 2^{nd} Ed, Prentice Hall, Upper Saddle River, NJ, 1999.
6. UTS, TK Solver User Guide, http://uts.com, Universal Technical Systems, Inc., Rockford, IL, 2006.
7. Young, W.C.; Budynas, R.G.: Roark's Formulas for Stress and Strain, 7-th Ed., McGraw-Hill, New York, NY, 2002.

